

2.34 hi2cin

72

*Syntaxe:***HI2CIN (variable,...)****HI2CIN location, (variable,...)****HI2CIN [newslave], (variable,...)****(seulement PICAXE X2)****HI2CIN [newslave], location, (variable,...)****(seulement PICAXE X2)**

- variable(s) dans laquelle la lecture de l'octet du périphérique i2c se positionne.
- location est une variable/constante spécifiant l'emplacement de départ de la lecture.
- newslave est une option pour une nouvelle adresse de commande.

Fonction:

Permet de lire les données du périphérique i2c et de la positionner dans une variable (ou plusieurs variables).

Information:

Les données sont lues à partir de l'adresse de départ indiquée, la lecture des autres données séquentielles est possible si le périphérique i2c le supporte.

L'emplacement doit être un octet ou un mot Définit par la commande i2csetup La commande i2csetup doit donc être Définit avant cette commande. Si l'on s'adresse à plusieurs périphériques i2c, il peut être nécessaire de changer l'adresse de l'esclave i2c en utilisant les options facultatives [newslave].

Si les périphériques i2c sont mal configurés ou que des données i2cslave erronées ont été utilisées, la valeur 255 (\$FF) est chargée dans chaque variable.

Exemple:

```
; Utilisation d'un module horloge DS1307
; ce module utilise le format BCD (voir utilisation de BCDTOASCII et BINTOASCII)
; le PICAXE est maître, le module DS1307 est l'esclave et utilise l'adresse i2c %11010000
```

```
hi2csetup i2cmaster, %11010000, i2cslow, i2cbyte ; adressage DS1307, mode
;slow en octet
```

debut:

```
hi2cin 0,(b0,b1,b2,b3,b4,b5,b6,b7) ; lecture et affichage de
; l'heure et de la date
debug ; affichage des valeurs sur l'écran de l'ordinateur
pause 2000 ; attente de 2 secondes
goto debut ; aller à debut, bouclage du programme
```

2.35 hi2cout

74

*Syntaxe:***HI2COUT** location, (variable,...)**HI2COUT** (variable,...)**HI2COUT** [newslave], location, (variable,...)

(seulement PICAXE X2)

HI2COUT [newslave], (variable,...)

(seulement PICAXE X2)

- variable contient l'octet à écrire.
- location est une variable/constante spécifiant l'emplacement de départ de l'écriture.
- newslave est une option pour une nouvelle adresse de commande.

Fonction:

Permet d'envoyer des données sur le bus i2c.

Information:

L'envoi d'une donnée s'effectue à partir de l'adresse de départ indiquée, l'envoi d'autres données séquentielles est possible si le périphérique i2c le supporte.

L'emplacement doit être un octet ou un mot Défini par la commande i2csetup La commande i2csetup doit donc être Défini avant cette commande. Si l'on s'adresse à plusieurs périphériques i2c, il peut être nécessaire de changer l'adresse de l'esclave i2c en utilisant les options facultatives [newslave].

Exemple:

```

; Utilisation d'un module horloge DS1307
; le PICAXE est maître, le module DS1307 est l'esclave et utilise l'adresse i2c %11010000
; écriture de l'heure à 11:59:00 et de la date au jeudi 17/10/13
; jeudi est le 05° jour de la semaine (dimanche le 01, samedi le 07)
symbol seconde = b0;
symbol minute = b1;
symbol heure = b2;
symbol jour = b3;
symbol date = b4;
symbol mois = b5;
symbol annee = b6;
symbol control = b7;
hi2csetup i2cmaster, %11010000, i2cslow, i2cbyte ; adressage DS1307
; mode slow en octet

let seconde = $00 ; positionne la variable seconde à 00
let minute = $59 ; positionne la variable minute à 59
let heure = $11 ; positionne la variable heure à 00
let jour = $05 ; positionne la variable jour au jeudi donc 05
let date = $17 ; positionne la variable date à 17
let mois = $10 ; positionne la variable mois à 10
let annee = $13 ; positionne la variable annee à 13
let control = $10 ; clignotement led du module horloge $10
; allumage led du module horloge $80
; extinction led du module horloge $00

hi2cout 0,(seconde,minute,heure,jour,date,mois,annee,control) ;chargement
;des infos dans le module horloge
end ; fin de la mise à l'heure du module horloge

```

2.36 hi2csetup

76

*Syntaxe:***HI2CSETUP OFF****HI2CSETUP I2CSLAVE, slaveaddress****HI2CSETUP I2CMASTER, slaveaddress, mode, addresslen**

Le mode master positionne le PICAXE en mode maître ce qui lui permet de contrôler le bus i2c.

Le mode slave positionne le PICAXE en mode esclave, dans ce mode il est commandé par un autre PICAXE en mode master.

- slaveaddress est l'adresse de l'esclave i2c.
- mode permet de choisir la vitesse de bus i2c : i2cfast(400kHz) ou i2cslow(100kHz). Utiliser i2cfast_8 ou i2cslow_8 pour les PICAXEs à 8MHZ.
- addresslen indique si la donnée est byte (i2cbyte) ou word (i2cword).

Fonction:

Permet de configurer le fonctionnement du bus i2c sur les broches du PICAXE.

Information:

L'utilisation des commandes i2c est abordée en détails dans la fiche technique du tutorial i2c.

Exemple:

```
; Utilisation d'un circuit i2c pcf8574
; le PICAXE est maître, le pcf8574 est l'esclave et utilise l'adresse i2c %01000000
```

```
symbol data = b0;
hi2csetup i2cmaster, %01000000, i2cslow, i2cbyte ; adressage pcf8574,
;mode slow en octet
```

debut:

```
let data = 240 ; charge data avec la valeur 240 soit %11110000
hi2cout (data) ; envoi de la donnée data
pause 500 ; attente de 0,5 seconde
let data = 15 ; charge data avec la valeur 15 soit %00001111
hi2cout (data) ; envoi de la donnée data
pause 500 ; attente de 0,5 seconde
goto debut ; aller à debut, bouclage du programme
```

2.37 hi2csetup - mode esclave (PICAXES X2 seulement)

76

Slave Address

L'adresse de l'esclave permet l'identification de celui-ci sur le bus i2c. C'est un nombre compris entre 1 et 127 à condition de respecter le codage sur les 7 bits de poids forts, bit7 à bit1 par exemple %1010000x .soit (160), le bit0 est utilisé pour indiquer le mode lecture/écriture et peut donc être ignoré.

Dans le cadre du protocole i2c, certaines adresses ont une signification particulière et ne sont pas recommandées car ils peuvent provoquer un comportement inattendu des appareils reliés au bus i2c.

Information:

Quand un PICAXE est en mode esclave, le PICAXE (ou un autre microcontrôleur) maître permet la lecture et l'écriture de l'esclave comme s'il s'agissait d'une mémoire EEPROM. Le transfert de donnée peut s'effectuer avec le scratchpad.

Les commandes qui désactivent les interruptions matérielles internes (par exemple SEROUT) peuvent affecter le fonctionnement. Voir l'annexe2 pour plus de détails sur les conflits possibles.

Lorsque le maître écrit dans la mémoire d'un PICAXE esclave, le « hi2cflag » est positionné et la dernière donnée écrite est positionnée dans la variable « hi2clast ». Il est possible de tester le bit « hi2cflag » ou en positionnant le setintflags, le programme du PICAXE pourra réagir en conséquence lorsqu'une opération d'écriture se produit. Attention, le hi2cflag doit être effacé par programme après utilisation.

L'exemple suivant indique comment utiliser 2 PICAXEs 28X1, l'un en maître et l'autre en esclave.

Exemple:

```
; Utilisation d'un PICAXE 28X1 utilisé en esclave
; le PICAXE esclave utilise l'adresse i2c %10100000
```

```
init:
    hi2csetup i2cslave,%10100000      ; indication de l'adresse i2c du PICAXE esclave

debut:
    if hi2cflag = 0 then debut        ; bouclage sur début tant que le flag est à 0
                                     ; permet d'attendre une information
    hi2cflag = 0                      ; raz du flag
    get hi2clast,b1                   ; dernier octet écrit dans la variable b1
    let outpins = b1                  ; positionne les pins de sortie avec la valeur de b1
    goto debut                         ; aller à debut, bouclage du programme
```

```
; Utilisation d'un PICAXE 28X1 en maître
```

```
init:
    hi2csetup i2cmaster,%10100000,i2cslow,i2cbyte ; initialisation i2c
                                                    en mode maitre

debut:
    inc b1                               ; incrémentation de la variable b1
    hi2cout 0,(b1)                       ; écriture de la donnée b1 dans l'esclave
    pause 500                             ; attente de 500ms
    goto debut                            ; aller à debut, bouclage du programme
```

2.38 hi2csetup - mode maître

78


Si vous utilisez un seul périphérique i2c esclave commandé par un PICAXE maître, cela nécessite une seule commande hi2csetup dans le programme. Après l'initialisation par le hi2csetup, il est possible d'accéder à l'esclave i2c avec les commandes hi2cin et hi2cout.

Si vous utilisez plusieurs esclaves, il est possible de changer l'adresse de l'esclave dans les commandes hi2cin et hi2cout avec [newslave].


Slave Address

L'adresse de l'esclave dépend du type de périphérique i2c (voir tableau ci-dessous). Pour les EEPROM 24LCXX, l'adresse i2c est organisée par une partie fixe (bit7 à bit 4) à 1010 et par une partie configurable (bit 3 à bit1) dépendant du câblage du circuit EEPROM. Cette astuce permet de pouvoir disposer jusqu'à 8 circuits EEPROM sur le bus i2c de %10100000 à %10101110.

Le bit0 est utilisé par le protocole i2c pour indiquer s'il s'agit d'une lecture ou d'une écriture de l'esclave, cependant les commandes PICAXE readi2c ou writei2c commandent le positionnement de ce bit0.



bits : A6 A5 A4 A3 A2 A1 A0 bit0
 adresse i2c : 1 0 1 0 0 0 0 0



bits : A6 A5 A4 A3 A2 A1 A0 bit0
 adresse i2c : 1 0 1 0 0 0 1 0

Les mémoires 24Cxx sont des mémoires à lecture et écriture électrique appelé EEPROM. Elles sont prévues pour supporter 1.000.000 cycles d'écriture et conserver l'information pendant au moins 40 ans....La broche 7 : WP permet d'autoriser l'écriture sur l'EEPROM.

La plupart des fiches techniques donnent l'adresse i2c en format 8 bits, si le format est sur 7 bits, il est nécessaire d'effectuer un décalage vers la gauche pour prendre en compte le bit0 de lecture/écriture.

Vitesse du bus i2c

La vitesse du bus i2c peut être sélectionnée en sélectionnant i2cfast(400kHz) ou i2cslow(100kHz). Utiliser i2cfast_8 ou i2cslow_8 pour les PICAXEs à 8MHZ ou i2cfast_16 ou i2cslow_16 pour les PICAXEs à 16MHZ.

Il est nécessaire d'utiliser la vitesse d'horloge la plus lente des périphériques du bus i2c, par exemple, il ne faut pas utiliser i2cfast si l'un des esclaves est à 100KHz (par exemple le module horloge DS1307)

Longueur de l'adresse

Les périphériques i2c peuvent utiliser des données sur 1 octet (mode i2cbyte) ou sur 2 octets (mode i2cword). Lors de l'utilisation de i2cword veillez à bien configurer les commandes hi2cin ou hi2cout pour prendre en compte la taille des données indiquées, sans précaution, un comportement erratique interviendra.

Device	Type	Slave	Speed	Mode
24LC01B	EE 128	%1010xxxx	i2cfast	i2cbyte
24LC02B	EE256	%1010xxxx	i2cfast	i2cbyte
24LC04B	EE 512	%1010xxbx	i2cfast	i2cbyte
24LC08B	EE 1kb	%1010xbbx	i2cfast	i2cbyte
24LC16B	EE 2kb	%1010bbbx	i2cfast	i2cbyte
24LC64	EE 8kb	%1010dddx	i2cfast	i2cword
24LC128	EE 16kb	%1010dddx	i2cfast	i2cword
24LC256	EE 32kb	%1010dddx	i2cfast	i2cword
24LC512	EE 64kb	%1010dddx	i2cfast	i2cword
DS1307	RTC	%1101000x	i2cslow	i2cbyte
MAX6953	5x7 LED	%101dddx	i2cfast	i2cbyte
AD5245	Digital Pot	%010110dx	i2cfast	i2cbyte
SRF08	Sonar	%1110000x	i2cfast	i2cbyte
AXE033	I2C LCD	\$C6	i2cslow	i2cbyte
CMPS03	Compass	%1100000x	i2cfast	i2cbyte
SPE030	Speech	%1100010x	i2cfast	i2cbyte

x : sans importance

b : bloc interne de sélection

d : dispositif de sélection (brochage)

La présence des résistances de pull-up de 4k7 est indispensable pour le fonctionnement du bus i2c :

